unit PrntPrev;





interface





uses


  SysUtils, WinTypes, WinProcs, Messages, Classes, Graphics, Controls,


  Forms, Dialogs, StdCtrls, ExtCtrls, Buttons, Mask, Print, CommDLG,


  Utility, DefAbort;





const


  PaperSizeNames = 64;      {Length of string of names of paper sizes}


  PaperBinsNames = 24;      {Length of string for names of paper bins}





type


  TWordArr = array[0..255] of Word;           {Word array, needed for type conversion}


  TPointArr = array[0..255] of TPOINT;        {TPOINT array, needed for type conversion}





  TPrinterOrientation = (poPortrait, poLandscape);    {Orientation, normally in Printers.}





  TPrintDlgOption = (pdAllPages, pdCollateCopies, pdDisablePrintToFile, pdHidePrintToFile, pdNoPageNums,


                         pdNoSelection, pdPageNums, pdPrintToFile, pdSelection, pdShowHelp);


  TPrintDlgOptions = set of TPrintDlgOption;  {Options for PrintDlg function}





  TMapMode = (mmText, mmLoMetric, mmHiMetric, mmLoEnglish, mmHiEnglish, mmTWIPs, mmIsotropic, mmAnisotropic);


  TAbortDlgMode = (dmNone, dmText, dmGauge);





  TPaperSize = record                         {Information needed for each size of paper supported}


    Size: Word;                               {Size, as used by API functions}


    Dimensions: TPOINT;                       {Dimensions in HiMetric (100/mm)}


    Name: array[0..PaperSizeNames] of Char;   {Name}


  end;


  TPaperSizeArr = array[0..255] of TPaperSize;{Array of paper sizes}


  PPaperSizeArr = ^TPaperSizeArr;             {Pointer to above}





  TPaperBins = record                         {Information needed for each paper bin supported}


    Bin:  Word;                               {Bin, as used by API functions}


    Name: array[0..PaperBinsNames] of Char;   {Name}


  end;


  TPaperBinsArr = array[0..255] of TPaperBins;{Array of bins}


  PPaperBinsArr = ^TPaperBinsArr;             {Pointer to above}





  TDevCapsFn = function( DeviceName, Port: PChar;           {DeviceCapabilities function}


                         Index: Word; Output: PChar; var DevMode: TDevMode): Longint;


  TRenderPageEvent = procedure(Page: Word; Canvas: TCanvas) of object;


  TAbortProcEvent = procedure(Page: Word) of object;





  TPrintPreview = class(TCustomPanel)


  private


    { Private declarations }


    FBevel: TBevel;                           {Border of main preview area}


    FScrollArea: TScrollBox;                  {Main preview area}


    FBackground: TPaintBox;                   {Background area (including borders)}


    FPrintableArea: TBitmap;                  {Bitmap to draw preview on, copied onto FBackground later}


    FScale: Double;                           {Scaling factor for page, just used for finding bitmap sizes}


    FIndent: TPoint;                          {Top left of page in preview window}


    FBGBit: TBitmap;                          {Bitmap used to fill in background}


    FControlPanel: TPanel;                    {Preview control panel}


    FUseZoom: Boolean;                        {Use the default zoom feature}


    FUsePageCtrls: Boolean;                   {Should the page controls be shown?}


    FDefCursorNum: Integer;                   {Default cursor number, is assigned to the magnify cursor}


    FCursor: TCursor;                         {Cursor used when on zoom area}


    FButFarLeft: TBitBtn;


    FButLeft: TBitBtn;


    FButRight: TBitBtn;


    FButFarRight: TBitBtn;


    FEditPageNum: TEdit;                      {Edit box for page number}


    FEditLabel: TLabel;                       {"Page:"}


    FZooming: Boolean;                        {Currently in the process of being zoomed, so don't redraw}


    FPreviewing: Boolean;                     {Same value as visible, which cannot be set}


    FAbort: Boolean;                          {Job is being aborted}


    FAbortDlgMode: TAbortDlgMode;             {Whether and what sort of abort dlg to show}


    FPrintDlg: TPrintDlg;                     {Global for persistence}


    FDeviceName: String;                      {Name of printer device}


    FOrientation: TPrinterOrientation;        {Orientation of paper}


    FPrintDlgOptions: TPrintDlgOptions;       {Set of options corresponding to TPrintDlg flags}


    FPaperSize: Integer;                      {Current paper size in API terms}


    FPaperSizeName: String;                   {Name of current paper size}


    FPaperSource: Integer;                    {Current paper bin in API terms}


    FPaperSourceName: String;                 {Name of current paper bin}


    FXdpi: Integer;                           {Xdpi of printer}


    FYdpi: Integer;                           {Ydpi of printer}


    FColour: Boolean;                         {True if printer is colour}


    FDuplex: Boolean;                         {True if printer can do duplex printing}


    FNumColours: Integer;                     {Number of colour printer can handle, always > 2}


    FIndentPixels: TPoint;                    {Distance to printable area, in pixels}


    FIndentHiMet: TPoint;                     {Distance to printable area, in mm/100}


    FPagePixels: TPoint;                      {Size of paper, in pixels}


    FPageHiMet: TPoint;                       {Size of paper, in mm/100}


    FPrintPixels: TPoint;                     {Printable area, in pixels}


    FPrintHiMet: TPoint;                      {Printable area, in mm/100}


    FPaperSizeArr: PPaperSizeArr;             {Pointer to array of PaperSize records}


    FPaperBinsArr: PPaperBinsArr;             {Pointer to array of PaperBin records}


    FNumSizes: Longint;                       {Number of entries in FPaperSizeArr^}


    FNumBins: Longint;                        {Number of entries in FPaperBinsArr^}


    FCurrentSize: Word;                       {Index to the PaperSizeArr for current paper size}


    FCurrentBin: Word;                        {Index to the PaperBinsArr for current paper bin}


    FPrtFileName: String;                     {Name of file for print to file, if needed will be set here}


    FCurZoom: Double;                         {Ratio of current size to full page size}


    FZoomPos: TPoint;                         {Location of centre of zoom in HiMet}


    FZoomFactor: Double;                      {Amount that zoom will toggle to/from is UseZoom is on}


    FMinPage: Word;


    FMaxPage: Word;


    FFromPage: Word;


    FToPage: Word;


    FCurrentPage: Word;                       {Page being previewed}


    FCurMapMode: TMapMode;                    {Map mode in operation, faked for preview, real for printer}


    FCopies: Word;


    FOnRenderPage: TRenderPageEvent;


    FOnChkAbort: TAbortProcEvent;


    FOnSelectionPages: TNotifyEvent;


    FOnMouseDown: TMouseEvent;


    procedure PrevPaint(Sender: TObject);


    procedure Loaded; override;


    procedure Resize; override;


    procedure PrevMouseDown(Sender: TObject; Button: TMouseButton; Shift: TShiftState; X, Y: Integer);


    procedure PrevMouseMove(Sender: TObject; Shift: TShiftState; X, Y: Integer);


    procedure GetDefaultPrinterInfo;


    procedure SetFScale;


    procedure ExtractPrinterInfo(PrintDlg: TPRINTDLG);


    procedure EmptyTPrintDlg(var PrintDlg: TPRINTDLG);


    procedure FillTPrintDlg(var PrintDlg: TPRINTDLG);


    procedure FillSizeBinsArrs(DriverName, DeviceName, Port: PChar; var DevMode: TDevMode);


    procedure StartPreviewRender;


    procedure EditPageNumKeyPress(Sender: TObject; var Key: Char);


    procedure EditPageNumChange(Sender: TObject);


    procedure FarLeftButClick(Sender: TObject);


    procedure LeftButClick(Sender: TObject);


    procedure RightButClick(Sender: TObject);


    procedure FarRightButClick(Sender: TObject);


    procedure CreateDefaultAbortDlg(JobName: String);


    procedure EndDefaultAbortDlg;


    procedure SetZoom(Pos: TPoint; Zoom: Double);


    function AbortProc(hdc: HDC; error: Integer): Boolean; export;


    function GetFileName: String;


    function CreateButton(Left: Integer; BmpName: PChar; ClickEvent: TNotifyEvent): TBitBtn;


    function GetPaperSize(Index: ShortInt): TPaperSize;


    function GetPaperBins(Index: ShortInt): TPaperBins;


    function GetZoomCursor: TCursor;


    procedure SetZoomCursor(Value: TCursor);


    procedure SetCurZoom(Value: Double);


    procedure SetZoomPos(Value: TPoint);


    procedure SetPreviewing(Value: Boolean);


    procedure SetCurrentPage(Value: Word);


    procedure SetAbort(Value: Boolean);


    procedure SetDefCursorNum(Value: Integer);


    procedure SetMinPage(Value: Word);


    procedure SetMaxPage(Value: Word);


    procedure SetFromPage(Value: Word);


    procedure SetToPage(Value: Word);


    procedure SetUsePageCtrls(Value: Boolean);


  protected


    { Protected declarations }


    procedure WMWinIniChange(var Message: TMessage); message CM_WININICHANGE;


  public


    { Public declarations }


    property Abort: Boolean read FAbort write SetAbort;


    property DeviceName: String read FDeviceName write FDeviceName;


    property Orientation: TPrinterOrientation read FOrientation write FOrientation;


    property PaperSize: Integer read FPaperSize write FPaperSize;


    property PaperSizeName: String read FPaperSizeName;


    property PaperSource: Integer read FPaperSource write FPaperSource;


    property PaperSourceName: String read FPaperSourceName;


    property PaperSizeArr[Index: ShortInt]: TPaperSize read GetPaperSize;


    property PaperBinsArr[Index: ShortInt]: TPaperBins read GetPaperBins;


    property NumSizes: Longint read FNumSizes;


    property NumBins: Longint read FNumBins;


    property Xdpi: Integer read FXdpi;


    property Ydpi: Integer read FYdpi;


    property Colour: Boolean read FColour;


    property Duplex: Boolean read FDuplex;


    property NumColours: Integer read FNumColours;


    property IndentPixels: TPoint read FIndentPixels;


    property IndentHiMet: TPoint read FIndentHiMet;


    property PagePixels: TPoint read FPagePixels;


    property PageHiMet: TPoint read FPageHiMet;


    property PrintPixels: TPoint read FPrintPixels;


    property PrintHiMet: TPoint read FPrintHiMet;


    property CurrentPage: Word read FCurrentPage write SetCurrentPage;


    property MapMode: TMapMode read FCurMapMode;


    property PrtFileName: String read FPrtFileName write FPrtFileName;


    property CurZoom: Double read FCurZoom write SetCurZoom;


    property ZoomPos: TPoint read FZoomPos write SetZoomPos;


    constructor Create(AOwner: TComponent); override;


    destructor Destroy; override;


    procedure SetupDlg;


    procedure PrintJobDlg;


    procedure StartPrintJob(JobName: String);


    procedure SetClipArea(Canvas: TCanvas; Rect: TRect);


    procedure UnSetClipArea(Canvas: TCanvas);


    function ChangeMapMode(Canvas: TCanvas; Mode: TMapMode; X, Y: Integer): TMapMode;


    function MouseToHiMet(MousePos: TPoint): TPoint;


  published


    { Published declarations}


    property UsePageCtrls: Boolean read FUsePageCtrls write SetUsePageCtrls default True;


    property UseZoom: Boolean read FUseZoom write FUseZoom default True;


    property ZoomFactor: Double read FZoomFactor write FZoomFactor;


    property DefCursorNum: Integer read FDefCursorNum write SetDefCursorNum default 5;


    property MinPage: Word read FMinPage write SetMinPage default $FFFF;


    property MaxPage: Word read FMaxPage write SetMaxPage default 1;


    property FromPage: Word read FFromPage write SetFromPage default 1;


    property ToPage: Word read FToPage write SetToPage default 1;


    property Copies: Word read FCopies write FCopies;


    property Options: TPrintDlgOptions read FPrintDlgOptions write FPrintDlgOptions default [];


    property Align default alClient;


    property Previewing: Boolean read FPreviewing write SetPreviewing default False;


    property UseAbortDlgMode: TAbortDlgMode read FAbortDlgMode write FAbortDlgMode;


    property ZoomCursor: TCursor read GetZoomCursor write SetZoomCursor;


    property OnRenderPage: TRenderPageEvent read FOnRenderPage write FOnRenderPage;


    property OnChkAbort: TAbortProcEvent read FOnChkAbort write FOnChkAbort;


    property OnResize;


    property OnSelectionPages: TNotifyEvent read FOnSelectionPages write FOnSelectionPages;


    property OnMouseDown: TMouseEvent read FOnMouseDown write FOnMouseDown;


  end;








procedure Register;





implementation





{$R PREVIEW.RES}





var


  GlobalAbortProc: TAbortProcEvent;


  GlobalPage: Word;


  GlobalAbort: Boolean;





constructor TPrintPreview.Create(AOwner: TComponent);


begin


  inherited Create(AOwner);


  FMinPage := 1;


  FMaxPage := $FFFF;


  FFromPage := 1;


  FToPage := 1;


  FPrintDlgOptions := [];


  Visible := False;


  FPreviewing := False;


  FZooming := False;


  FPaperSizeArr := nil;


  FPaperBinsArr := nil;


  FPrtFileName := '';


  FDefCursorNum := 5;


  FUseZoom := True;


  FUsePageCtrls := True;


  FCurZoom := 1.0;


  FZoomFactor := 3.0;


  FScale := 0.0;





  Align := alClient;


  BevelOuter := bvNone;


  BevelInner := bvNone;





  FControlPanel := TPanel.Create(Self);


  FControlPanel.Parent := Self;


  FControlPanel.Align := alBottom;


  FControlPanel.Height := 18;


  FControlPanel.BevelOuter := bvNone;





  FBGBit := TBitmap.Create;


  FBGBit.Handle := LoadBitmap(HInstance, 'CHECK');





  FBevel := TBevel.Create(Self);


  FBevel.Parent := Self;


  FBevel.Align := alClient;


  FBevel.Style := bsLowered;


  FBevel.Visible := True;





  FScrollArea := TScrollBox.Create(Self);


  FScrollArea.Parent := Self;


  FScrollArea.Align := alClient;


  FScrollArea.Visible := True;


  FScrollArea.BorderStyle := bsSingle;


  FScrollArea.Color := clSilver;


  FScrollArea.AutoScroll := True;





  FBackground := TPaintBox.Create(FScrollArea);


  FBackground.Parent := FScrollArea;


  FBackground.Visible := True;





  FPrintableArea := TBitmap.Create;





  FButFarLeft := CreateButton(0, 'FARLEFT', FarLeftButClick);


  FButLeft := CreateButton(16, 'LEFT', LeftButClick);


  FButRight := CreateButton(113, 'RIGHT', RightButClick);


  FButFarRight := CreateButton(129, 'FARRIGHT', FarRightButClick);





  FEditPageNum := TEdit.Create(FControlPanel);


  FEditPageNum.Parent := FControlPanel;


  FEditPageNum.AutoSize := False;


  FEditPageNum.Left := 75;


  FEditPageNum.Height := 17;


  FEditPageNum.Top := 1;


  FEditPageNum.Width := 35;


  FEditPageNum.Font.Name := 'Arial';


  FEditPageNum.Font.Height := 16;


  FEditPageNum.Ctl3D := False;


  FEditPageNum.ParentCtl3D := False;


  FEditPageNum.OnKeyPress := EditPageNumKeyPress;


  FEditPageNum.OnChange := EditPageNumChange;


  FEditPageNum.OnExit := EditPageNumChange;





  FEditLabel := TLabel.Create(Self);


  FEditLabel.Parent := FControlPanel;


  FEditLabel.Top := 1;


  FEditLabel.Left := 37;


  FEditLabel.Font.Name := 'Arial';


  FEditLabel.Font.Height := 16;


  FEditLabel.Caption := 'Page:';


end;





destructor TPrintPreview.Destroy;


begin


  if FPaperSizeArr <> nil then


    FreeMem(FPaperSizeArr, NumSizes * sizeof(TPaperSize));


  if FPaperBinsArr <> nil then


    FreeMem(FPaperBinsArr, NumBins * sizeof(TPaperBins));


  if FBGBit.Handle <> 0 then


    DeleteObject(FBGBit.Handle);


  GlobalFree(FPrintDlg.hDevMode);


  GlobalFree(FPrintDlg.hDevNames);


  inherited Destroy;


end;





procedure TPrintPreview.Loaded;


begin


  inherited Loaded;


  GetDefaultPrinterInfo;


  Screen.Cursors[DefCursorNum] := LoadCursor(HInstance, 'MAGNIFY');


  FControlPanel.Visible := FUsePageCtrls;


  FBackground.OnPaint := PrevPaint;


  FBackground.OnMouseDown := PrevMouseDown;


  FBackground.OnMouseMove := PrevMouseMove;


  Resize;


end;





function TPrintPreview.GetPaperSize(Index: ShortInt): TPaperSize;


var


  RetVal: TPaperSize;


begin


  if (NumSizes = 0) or (Index > NumSizes - 1) or (Index < 0) then


  begin


    RetVal.Size := DMPAPER_USER	;


    RetVal.Dimensions.X := 0;


    RetVal.Dimensions.Y := 0;


    StrPCopy(RetVal.Name, 'None');


  end


  else


  begin


    RetVal := FPaperSizeArr^[Index];


  end;


  GetPaperSize := RetVal;


end;





function TPrintPreview.GetPaperBins(Index: ShortInt): TPaperBins;


var


  RetVal: TPaperBins;


begin


  if (NumBins = 0) or (Index > NumBins - 1) or (Index < 0) then


  begin


    RetVal.Bin := DMBIN_AUTO;


    StrPCopy(RetVal.Name, 'None');


  end


  else


  begin


    RetVal := FPaperBinsArr^[Index];


  end;


  GetPaperBins := RetVal;


end;





procedure TPrintPreview.SetPreviewing(Value: Boolean);


begin


  if Value <> FPreviewing then


  begin


    FPreviewing := Value;


    Visible := Value;


    if (CurrentPage < FromPage) or (CurrentPage > ToPage) then


      CurrentPage := FromPage


    else


      CurrentPage := CurrentPage;


  end;


end;





procedure TPrintPreview.SetCurrentPage(Value: Word);


begin


  if Value <> FCurrentPage then


    FEditPageNum.Text := IntToStr(Value);


  StartPreviewRender;


  FBackground.Invalidate;


end;





procedure TPrintPreview.SetAbort(Value: Boolean);


begin


  if Value <> FAbort then


  begin


    FAbort := Value;


    GlobalAbort := Value;


  end;


end;





function TPrintPreview.GetZoomCursor: TCursor;


begin


  GetZoomCursor := FCursor;


end;





procedure TPrintPreview.SetZoomCursor(Value: TCursor);


begin


  if FCursor <> Value then


    FCursor := Value;


end;





procedure TPrintPreview.SetCurZoom(Value: Double);


begin


  if (Value <> FCurZoom) and (Value <> 0) then


  begin


    FCurZoom := Value;


    FScale := FScale / CurZoom;


    Resize;


  end;


end;





procedure TPrintPreview.SetZoomPos(Value: TPoint);


begin


  if (Value.X <> FZoomPos.X) or (Value.Y <> FZoomPos.Y) then


    FZoomPos := Value;


  Resize;


end;





procedure TPrintPreview.SetZoom(Pos: TPoint; Zoom: Double);


begin


  FZooming := True;


  if (Zoom <> FCurZoom) and (Zoom <> 0) then


  begin


    FCurZoom := Zoom;


    FScale := FScale / CurZoom;


  end;


  if (Pos.X <> FZoomPos.X) or (Pos.Y <> FZoomPos.Y) then


  begin


    FZoomPos := Pos;


  end;


  if FCurZoom = 1 then


  begin


    FScrollArea.HorzScrollBar.Position := 0;


    FScrollArea.VertScrollBar.Position := 0;


  end;


  Resize;


  if FScale <> 0 then


  begin


    FScrollArea.HorzScrollBar.Position := Integer(Round(Pos.X / FScale)) + FIndent.X - FScrollArea.Width div 2;


    FScrollArea.VertScrollBar.Position := Integer(Round(Pos.Y / FScale)) + FIndent.Y - FScrollArea.Height div 2;


  end;


  FZooming := False;


  FBackground.Invalidate;


end;





procedure TPrintPreview.SetMinPage(Value: Word);


begin


  if Value <> FMinPage then


  begin


    FMinPage := Value;


    if FromPage < FMinPage then


      FromPage := FMinPage;


  end;


end;





procedure TPrintPreview.SetMaxPage(Value: Word);


begin


  if Value <> FMaxPage then


  begin


    FMaxPage := Value;


    if ToPage > FMaxPage then


      ToPage := FMaxPage;


  end;


end;





procedure TPrintPreview.SetFromPage(Value: Word);


begin


  if (Value <> FFromPage) and (Value >= MinPage) then


    FFromPage := Value;


end;





procedure TPrintPreview.SetToPage(Value: Word);


begin


  if (Value <> FToPage) and (Value <= MaxPage) then


    FToPage := Value;


end;





procedure TPrintPreview.SetDefCursorNum(Value: Integer);


begin


  if ((csLoading in ComponentState) or (csDesigning in ComponentState)) and (Value <> FDefCursorNum) then


    FDefCursorNum := Value;


end;





procedure TPrintPreview.SetUsePageCtrls(Value: Boolean);


begin


  if Value <> FUsePageCtrls then


  begin


    FUsePageCtrls := Value;


    FControlPanel.Visible := Value;


  end;


end;





function TPrintPreview.ChangeMapMode(Canvas: TCanvas; Mode: TMapMode; X, Y: Integer): TMapMode;


var


  DoubleX, DoubleY: Double;


begin


  if Assigned(FPrintableArea) and (Canvas = FPrintableArea.Canvas) then


  begin


    {This is the bitmap for the preview}


    FCurMapMode := Mode;


    SetMapMode(Canvas.Handle, MM_ANISOTROPIC);


    SetViewportExt( Canvas.Handle, FPrintableArea.Width, FPrintableArea.Height);


    case Mode of


     mmText:


      begin


        SetWindowExt( Canvas.Handle, FPrintPixels.X, FPrintPixels.Y);


      end;


     mmLoMetric:


      begin


        SetWindowExt( Canvas.Handle, FPrintHiMet.X div 10, FPrintHiMet.Y div -10);


      end;


     mmHiMetric:


      begin


        SetWindowExt( Canvas.Handle, FPrintHiMet.X , 0 - FPrintHiMet.Y);


      end;


     mmLoEnglish:


      begin


        SetWindowExt( Canvas.Handle,


                      Integer(Round(FPrintHiMet.X / 25.4)),


                      Integer(Round(FPrintHiMet.Y / -25.4)));


      end;


     mmHiEnglish:


      begin


        SetWindowExt( Canvas.Handle,


                      Integer(Round(FPrintHiMet.X / 2.54)),


                      Integer(Round(FPrintHiMet.Y / -2.54)));


      end;


     mmTWIPs:


      begin


        SetWindowExt( Canvas.Handle,


                      Integer(Round(FPrintHiMet.X / 1.76)),


                      Integer(Round(FPrintHiMet.Y / -1.76)));


      end;


     mmIsotropic:


      begin


        DoubleX := X;


        DoubleY := Y;


        if (FPrintHiMet.X / DoubleX) < (FPrintHiMet.Y / DoubleY) then


          SetWindowExt( Canvas.Handle, X, Integer(Round(DoubleY * FPrintHiMet.Y / FPrintHiMet.X)))


        else


          SetWindowExt( Canvas.Handle, Integer(Round(DoubleX * FPrintHiMet.X / FPrintHiMet.Y)), Y);


      end;


     mmAnisotropic:


        SetWindowExt( Canvas.Handle, X, Y);


    else


      FCurMapMode := mmHiMetric;


      SetWindowExt( Canvas.Handle, FPrintHiMet.X, FPrintHiMet.Y);


    end;


    ChangeMapMode := FCurMapMode;


  end


  else


  begin


    ChangeMapMode := Mode;


    case Mode of


     mmText:      SetMapMode(Canvas.Handle, MM_TEXT);


     mmLoMetric:  SetMapMode(Canvas.Handle, MM_LOMETRIC);


     mmHiMetric:  SetMapMode(Canvas.Handle, MM_HIMETRIC);


     mmLoEnglish: SetMapMode(Canvas.Handle, MM_LOENGLISH);


     mmHiEnglish: SetMapMode(Canvas.Handle, MM_HIENGLISH);


     mmTWIPs:     SetMapMode(Canvas.Handle, MM_TWIPS);


     mmIsotropic:


      begin


        SetMapMode(Canvas.Handle, MM_ISOTROPIC);


        SetWindowExt(Canvas.Handle, X, Y);


        SetViewportExt(Canvas.Handle, FPrintPixels.X - 1, FPrintPixels.Y - 1);


      end;


     mmAnisotropic:


      begin


        SetMapMode(Canvas.Handle, MM_ANISOTROPIC);


        SetWindowExt(Canvas.Handle, X, Y);


        SetViewportExt(Canvas.Handle, FPrintPixels.X - 1, FPrintPixels.Y - 1);


      end;


    else


      SetMapMode(Canvas.Handle, MM_HIMETRIC);


      ChangeMapMode := mmHiMetric;


    end;


  end;


end;





procedure TPrintPreview.SetClipArea(Canvas: TCanvas; Rect: TRect);


var


  ClipRgn: HRgn;


  Points: array[0..1] of TPoint;


begin


  Points[0].X := Rect.Left;


  Points[0].Y := Rect.Top;


  Points[1].X := Rect.Right;


  Points[1].Y := Rect.Bottom;


  LPtoDP(Canvas.Handle, Points, 2);


  ClipRgn := CreateRectRgn(Points[0].X, Points[0].Y, Points[1].X, Points[1].Y);


  SelectClipRgn(Canvas.Handle, ClipRgn);


  DeleteObject(ClipRgn);


end;





procedure TPrintPreview.UnSetClipArea(Canvas: TCanvas);


begin


  SelectClipRgn(Canvas.Handle, HRgn(nil));


end;





procedure TPrintPreview.EditPageNumKeyPress(Sender: TObject; var Key: Char);


begin


  if ((Ord(Key) < Ord('0')) or (Ord(Key) > Ord('9'))) and (Ord(Key) > 31) then


    Key := Chr(0);


end;





procedure TPrintPreview.EditPageNumChange(Sender: TObject);


begin


  if not FEditPageNum.Focused then


  begin


    try


      FCurrentPage := StrToInt(FEditPageNum.Text);


    except


      FCurrentPage := FromPage;


    end;





    if FCurrentPage > MaxPage then


      FCurrentPage := MaxPage;


    if FCurrentPage < MinPage then


      FCurrentPage := MinPage;





    if FEditPageNum.Text = IntToStr(FCurrentPage) then


    begin


      StartPReviewRender;


      FBackground.Invalidate;


    end


    else


      FEditPageNum.Text := IntToStr(FCurrentPage);


  end;


end;





procedure TPrintPreview.FarLeftButClick(Sender: TObject);


begin


  CurrentPage := FMinPage;


end;





procedure TPrintPreview.LeftButClick(Sender: TObject);


begin


  if FCurrentPage > FMinPage then


    CurrentPage := FCurrentPage - 1;


end;





procedure TPrintPreview.RightButClick(Sender: TObject);


begin


  if FCurrentPage < FMaxPage then


    CurrentPage := FCurrentPage + 1;


end;





procedure TPrintPreview.FarRightButClick(Sender: TObject);


begin


  CurrentPage := FMaxPage;


end;





procedure TPrintPreview.Resize;


var


  OldSize: TPoint;


  Factor: Double;


begin


  FBackground.SetBounds( 0, 0, 0, 0);


  inherited Resize;


  SetFScale;


  if FScale <> 0 then


  begin


    FIndent.X := MaxInteger(Integer(Round((FScrollArea.Width  - (PageHiMet.X / FScale)) / 2)), 8);


    FIndent.Y := MaxInteger(Integer(Round((FScrollArea.Height  - (PageHiMet.Y / FScale)) / 2)), 8);


    OldSize.X := FPrintableArea.Width;


    OldSize.Y := FPrintableArea.Height;


    FPrintableArea.Monochrome := not Colour;


    FPrintableArea.Width := Integer(Round(PrintHiMet.X / FScale));


    FPrintableArea.Height := Integer(Round(PrintHiMet.Y / FScale));


    if (OldSize.X <> FPrintableArea.Width) or (OldSize.Y <> FPrintableArea.Height) then


    begin


      StartPreviewRender;


    end;


    FBackground.SetBounds( 0, 0,


                           MaxInteger(Integer(Round(PageHiMet.X / FScale)) + 14, FScrollArea.ClientWidth),


                           MaxInteger(Integer(Round(PageHiMet.Y / FScale)) + 14, FScrollArea.ClientHeight));


  end;


  FBackground.Refresh;


end;





procedure TPrintPreview.PrevMouseDown(Sender: TObject; Button: TMouseButton; Shift: TShiftState; X, Y: Integer);


var


  ClickPos: TPoint;


  IndentX, IndentY: Integer;


begin


  if Assigned(FOnMouseDown) then


    OnMouseDown(Self, Button, Shift, X, Y);


  if UseZoom and (FScale <> 0) then


  begin


    IndentX := FIndent.X + Integer(Round(FIndentHiMet.X / FScale));


    IndentY := FIndent.Y + Integer(Round(FIndentHiMet.Y / FScale));


    if (X > IndentX) and (X < IndentX + FPrintableArea.Width) and


       (Y > IndentY) and (Y < IndentY + FPrintableArea.Height) then


    begin


      ClickPos := MouseToHiMet(Point(X, Y));


      if CurZoom = 1.0 then


        SetZoom(ClickPos, ZoomFactor)


      else


        SetZoom(ClickPos, 1.0);


    end;


  end;


end;





procedure TPrintPreview.PrevMouseMove(Sender: TObject; Shift: TShiftState; X, Y: Integer);


var


  IndentX, IndentY: Integer;


begin


  if UseZoom and (FScale <> 0) then


  begin


    IndentX := FIndent.X + Integer(Round(FIndentHiMet.X / FScale));


    IndentY := FIndent.Y + Integer(Round(FIndentHiMet.Y / FScale));


    if (X > IndentX) and (X < IndentX + FPrintableArea.Width) and


       (Y > IndentY) and (Y < IndentY + FPrintableArea.Height) then


      FBackground.Cursor := ZoomCursor


    else


      FBackground.Cursor := crDefault;


  end;


end;





function TPrintPreview.CreateButton(Left: Integer; BmpName: PChar; ClickEvent: TNotifyEvent): TBitBtn;


var


  Button: TBitBtn;


begin


  Button := TBitBtn.Create(FControlPanel);


  Button.Parent := FControlPanel;


  Button.Left := Left;


  Button.Top := 1;


  Button.Width := 17;


  Button.Height := 17;


  Button.Enabled := True;


  Button.Glyph.Handle := LoadBitmap(HInstance, BmpName);


  Button.NumGlyphs := 1;


  Button.Invalidate;


  Button.OnClick := ClickEvent;


end;





procedure TPrintPreview.GetDefaultPrinterInfo;


begin


  FPrintDlg.hDevMode := THandle(nil);


  FPrintDlg.hDevNames := THandle(nil);


  FPrintDlg.hDC := HDC(nil);


  FPrintDlg.Flags := PD_RETURNIC or PD_RETURNDEFAULT;


  FillTPrintDlg(FPrintDlg);


  PrintDlg(FPrintDlg);


  ExtractPrinterInfo(FPrintDlg);


  DeleteDC(FPrintDlg.hDC);


end;





procedure TPrintPreview.SetupDlg;


begin


  FPrintDlg.hDC := HDC(nil);


  FPrintDlg.Flags := PD_RETURNIC or PD_PRINTSETUP;


  FillTPrintDlg(FPrintDlg);


  if BOOL(PrintDlg(FPrintDlg)) then


  begin


    ExtractPrinterInfo(FPrintDlg);


    DeleteDC(FPrintDlg.hDC);


  end;


  Resize;


end;





procedure TPrintPreview.PrintJobDlg;


begin


  FPrintDlg.hDC := HDC(nil);


  FPrintDlg.Flags := PD_RETURNIC;


  FillTPrintDlg(FPrintDlg);


  if BOOL(PrintDlg(FPrintDlg)) then


  begin


    ExtractPrinterInfo(FPrintDlg);


    DeleteDC(FPrintDlg.hDC);


  end;


  Resize;


end;





procedure TPrintPreview.StartPrintJob(JobName: String);


var


  DocInfo: TDocInfo;


  PrinterDC: HDC;


  Filename: array[0..128] of char;


  JobPChars: array[0..128] of char;


  TempName: String;


  Page: Word;


  pDevMode: ^TDEVMODE;


  pDevNames: ^TDEVNAMES;


  PrtCanv: TCanvas;


  AbortProcInst: TFarProc;


begin


  GlobalAbort := False;


  if (FPrintDlg.hDevMode <> THandle(nil)) and (FPrintDlg.hDevNames <> THandle(nil)) then


  try


    pDevNames := GlobalLock(FPrintDlg.hDevNames);


    pDevMode := GlobalLock(FPrintDlg.hDevMode);





    TempName := FPrtFileName;


    if BOOL(FPrintDlg.Flags and PD_PRINTTOFILE) and (FPrtFileName = '') then


      TempName := GetFileName;





    if (not BOOL(FPrintDlg.Flags and PD_PRINTTOFILE)) or (TempName <> '') then


    begin


      try


        PrinterDC := CreateDC( PChar(pDevNames) + pDevNames^.wDriverOffset,


                               PChar(pDevNames) + pDevNames^.wDeviceOffset,


                               PChar(pDevNames) + pDevNames^.wOutputOffset, pDevMode);


        PrtCanv := TCanvas.Create;


        PrtCanv.Handle := PrinterDC;





        if Assigned(FOnChkAbort) then


          GlobalAbortProc := FOnChkAbort;





        StrPCopy(JobPChars, JobName);


        DocInfo.cbSize := sizeof(TDOCINFO);


        DocInfo.lpszDocName := JobPChars;


        if TempName <> '' then


        begin


          StrPCopy(Filename, TempName);


          DocInfo.lpszOutput := Filename;


        end


        else


          DocInfo.lpszOutput := nil;





        StartDoc(PrinterDC, DocInfo);


        if UseAbortDlgMode <> dmNone then


          CreateDefaultAbortDlg(JobName);


        EnableWindow(GetParentForm(Self).Handle, False);


        AbortProcInst := MakeProcInstance(@TPrintPreview.AbortProc, HInstance);


        SetAbortProc(PrinterDC, TAbortProc(AbortProcInst));


        for Page := FromPage to ToPage do


        begin


          GlobalPage := Page;


          StartPage(PrinterDC);


          if Assigned(FOnRenderPage) and not GlobalAbort then


          begin


            ChangeMapMode(PrtCanv, mmHiMetric, 0, 0);


            OnRenderPage(Page, PrtCanv);


          end;


          EndPage(PrinterDC);


        end;


        EndDoc(PrinterDC);


        EnableWindow(GetParentForm(Self).Handle, True);


        if UseAbortDlgMode <> dmNone then


          EndDefaultAbortDlg;


        FreeProcInstance(AbortProcInst);


      finally


        DeleteDC(PrinterDC);


        PrtCanv.Free;


      end;


    end;


  finally


    GlobalUnlock(FPrintDlg.hDevMode);


    GlobalUnlock(FPrintDlg.hDevNames);


  end;


end;





procedure TPrintPreview.ExtractPrinterInfo(PrintDlg: TPRINTDLG);


var


  pDevMode: ^TDEVMODE;


  pDevNames: ^TDEVNAMES;


  Loopy: Integer;


  TempPoint: TPOINT;


begin


  EmptyTPrintDlg(PrintDlg);





  if PrintDlg.hDevNames = THandle(nil) then


    MessageDlg('No TDEVNAMES setup!', mtError, [mbCancel], 0)


  else


  begin


    pDevNames := GlobalLock(PrintDlg.hDevNames);


    FDeviceName := StrPas(PChar(pDevNames) + pDevNames^.wDeviceOffset);





    if PrintDlg.hDevMode = THandle(nil) then


      MessageDlg('No TDEVMODE setup!', mtError, [mbCancel], 0)


    else


    begin


      pDevMode := GlobalLock(PrintDlg.hDevMode);


      if BOOL(pDevMode^.dmFields and DM_PAPERSIZE) then


        FPaperSize := pDevMode^.dmPaperSize;





      FillSizeBinsArrs( PChar(pDevNames) + pDevNames^.wDriverOffset,


                        PChar(pDevNames) + pDevNames^.wDeviceOffset,


                        PChar(pDevNames) + pDevNames^.wOutputOffset,


                        pDevMode^);





      FPaperSize := DMPAPER_USER;


      for Loopy := 0 to NumSizes - 1 do


        if PaperSizeArr[Loopy].Size = pDevMode^.dmPaperSize then


        begin


          FCurrentSize := Loopy;


          FPaperSize := PaperSizeArr[Loopy].Size;


          FPaperSizeName := StrPas(PaperSizeArr[Loopy].Name);


         end;





      if FPaperSize = DMPAPER_USER then


        if BOOL(pDevMode^.dmFields and DM_PAPERLENGTH) and BOOL(pDevMode^.dmFields and DM_PAPERWIDTH) then


        begin


          FPaperSizeName := 'User Defined Size';


        end;





      FCurrentBin := NumBins + 1;


      for Loopy := 0 to NumBins - 1 do


        if PaperBinsArr[Loopy].Bin = pDevMode^.dmDefaultSource then


        begin


          FCurrentBin := Loopy;


          FPaperSource := PaperBinsArr[Loopy].Bin;


          FPaperSourceName := StrPas(PaperBinsArr[Loopy].Name);


        end;





      if FCurrentBin = NumBins + 1 then


      begin


        FPaperSource := DMBIN_AUTO;


        FPaperSourceName := 'Unknown';


      end;





      FOrientation := poPortrait;


      if BOOL(pDevMode^.dmFields and DM_ORIENTATION) and (pDevMode^.dmOrientation = DMORIENT_LANDSCAPE) then


        FOrientation := poLandscape;





      FDuplex := False;


      if BOOL(pDevMode^.dmFields and DM_DUPLEX) and (pDevMode^.dmDuplex <> DMDUP_SIMPLEX) then


        FDuplex := True;





      FColour := False;


      FNumColours := 2;


      if pDevMode^.dmColor = DMCOLOR_COLOR then


      begin


        FColour := True;


        FNumColours := GetDeviceCaps(PrintDlg.hDC, NUMCOLORS);


      end;





      SetMapMode(PrintDlg.hDC, MM_HIMETRIC);





      Escape(PrintDlg.hDC, GETPRINTINGOFFSET, 0, nil, @FIndentPixels);


      FIndentHiMet := FIndentPixels;


      DPtoLP(PrintDlg.hDC, FIndentHiMet, 1);


      FIndentHiMet.Y := 0 - FIndentHiMet.Y;





      Escape(PrintDlg.hDC, GETPHYSPAGESIZE, 0, nil, @FPagePixels);


      FPageHiMet := FPagePixels;


      DPtoLP(PrintDlg.hDC, FPageHiMet, 1);


      FPageHiMet.Y := 0 - FPageHiMet.Y;





      FPrintPixels.X := GetDeviceCaps(PrintDlg.hDC, HORZRES);


      FPrintPixels.Y := GetDeviceCaps(PrintDlg.hDC, VERTRES);


      FPrintHiMet := FPrintPixels;


      DPtoLP(PrintDlg.hDC, FPrintHiMet, 1);


      FPrintHiMet.Y := 0 - FPrintHiMet.Y;





      FXdpi := GetDeviceCaps(PrintDlg.hDC, LOGPIXELSX);


      FYdpi := GetDeviceCaps(PrintDlg.hDC, LOGPIXELSY);





      GlobalUnlock(PrintDlg.hDevMode);


    end;


    GlobalUnlock(PrintDlg.hDevNames);


  end;


  SetFScale;


end;





{Fills the TPRINTDLG with the data from the properties}


procedure TPrintPreview.FillTPrintDlg(var PrintDlg: TPRINTDLG);


begin


  PrintDlg.lStructSize := sizeof(TPRINTDLG);


  PrintDlg.hWndOwner := Application.Handle;


  PrintDlg.nFromPage := FromPage;


  PrintDlg.nToPage := ToPage;


  PrintDlg.nMinPage := MinPage;


  PrintDlg.nMaxPage := MaxPage;


  PrintDlg.nCopies := Copies;


  if pdAllPages in Options then


    PrintDlg.Flags := PrintDlg.Flags or PD_ALLPAGES;


  if pdCollateCopies in Options then


    PrintDlg.Flags := PrintDlg.Flags or PD_COLLATE;


  if pdDisablePrintToFile in Options then


    PrintDlg.Flags := PrintDlg.Flags or PD_DISABLEPRINTTOFILE;


  if pdHidePrintToFile in Options then


    PrintDlg.Flags := PrintDlg.Flags or PD_HIDEPRINTTOFILE;


  if pdNoPageNums in Options then


    PrintDlg.Flags := PrintDlg.Flags or PD_NOPAGENUMS;


  if pdNoSelection in Options then


    PrintDlg.Flags := PrintDlg.Flags or PD_NOSELECTION;


  if pdPageNums in Options then


    PrintDlg.Flags := PrintDlg.Flags or PD_PAGENUMS;


  if pdPrintToFile in Options then


    PrintDlg.Flags := PrintDlg.Flags or PD_PRINTTOFILE;


  if pdSelection in Options then


    PrintDlg.Flags := PrintDlg.Flags or PD_SELECTION;


  if pdShowHelp in Options then


    PrintDlg.Flags := PrintDlg.Flags or PD_SHOWHELP;


end;





{R3emoves the flags and page nums from the TPRINTDLG with the data from the properties}


procedure TPrintPreview.EmptyTPrintDlg(var PrintDlg: TPRINTDLG);


begin


  FFromPage := MinPage;


  FToPage := MaxPage;


  if Bool(PrintDlg.Flags and PD_PAGENUMS) then


  begin


    FFromPage := PrintDlg.nFromPage;


    FToPage := PrintDlg.nToPage;


  end;


  if Bool(PrintDlg.Flags and PD_SELECTION) and Assigned(FOnSelectionPages) then


    FOnSelectionPages(Self);


  Copies := PrintDlg.nCopies;





  Options := [];


  if Bool(PrintDlg.Flags and PD_ALLPAGES) then


    Options := Options + [pdAllPages];


  if Bool(PrintDlg.Flags and PD_COLLATE) then


    Options := Options + [pdCollateCopies];


  if Bool(PrintDlg.Flags and PD_DISABLEPRINTTOFILE) then


    Options := Options + [pdDisablePrintToFile];


  if Bool(PrintDlg.Flags and PD_HIDEPRINTTOFILE) then


    Options := Options + [pdHidePrintToFile];


  if Bool(PrintDlg.Flags and PD_NOPAGENUMS) then


    Options := Options + [pdNoPageNums];


  if Bool(PrintDlg.Flags and PD_NOSELECTION) then


    Options := Options + [pdNoSelection];


  if Bool(PrintDlg.Flags and PD_PAGENUMS) then


    Options := Options + [pdPageNums];


  if Bool(PrintDlg.Flags and PD_PRINTTOFILE) then


    Options := Options + [pdPrintToFile];


  if Bool(PrintDlg.Flags and PD_SELECTION) then


    Options := Options + [pdSelection];


  if Bool(PrintDlg.Flags and PD_SHOWHELP) then


    Options := Options + [pdShowHelp];


end;





{Collects information regarding possible paper size and bins}


procedure TPrintPreview.FillSizeBinsArrs( DriverName, DeviceName, Port: PChar; var DevMode: TDevMode);


var


  LibHandle: THandle;


  Proc: TFarProc;


  DeviceCapabilities: TDevCapsFn;


  Loopy: Word;


  PaperSizes, PaperDims, PaperNames, Bins, BinNames: Pointer;


  OldNumSizes, OldNumBins: LongInt;


begin


  LibHandle := LoadLibrary(DriverName);


  Proc := GetProcAddress(LibHandle, 'DeviceCapabilities');


  DeviceCapabilities := TDevCapsFn(Proc);


  {Memory allocation for size and bins arrays}


  {Three options:  Not yet allocated, allocated of same size, or allocated of other size}


  {Paper sizes first}


  OldNumSizes := FNumSizes;


  FNumSizes := DeviceCapabilities(DeviceName, Port, DC_PAPERS, PChar(nil), DevMode);


  if FPaperSizeArr = nil then


    FPaperSizeArr := AllocMem(NumSizes * sizeof(TPaperSize))


  else if OldNumSizes <> FNumSizes then


    FPaperSizeArr := ReallocMem(FPaperSizeArr, OldNumSizes * sizeof(TPaperSize), NumSizes * sizeof(TPaperSize));





  {Now bins}


  OldNumBins := FNumBins;


  FNumBins := DeviceCapabilities(DeviceName, Port, DC_BINS, PChar(nil), DevMode);


  if FPaperBinsArr = nil then


    FPaperBinsArr := AllocMem(NumBins * sizeof(TPaperBins))


  else if OldNumBins <> FNumBins then


    FPaperBinsArr := ReallocMem(FPaperBinsArr, OldNumBins * sizeof(TPaperBins), NumSizes * sizeof(TPaperBins));





  PaperSizes := AllocMem(NumSizes * sizeof(Word));


  PaperDims := AllocMem(NumSizes * sizeof(TPOINT));


  PaperNames := AllocMem(NumSizes * PaperSizeNames * sizeof(char));


  Bins := AllocMem(NumBins * sizeof(Word));


  BinNames := AllocMem(NumBins * PaperBinsNames * sizeof(char));





  DeviceCapabilities(DeviceName, Port, DC_PAPERS, PChar(PaperSizes), DevMode);


  DeviceCapabilities(DeviceName, Port, DC_PAPERSIZE, PChar(PaperDims), DevMode);


  DeviceCapabilities(DeviceName, Port, DC_PAPERNAMES, PChar(PaperNames), DevMode);


  DeviceCapabilities(DeviceName, Port, DC_BINS, PChar(Bins), DevMode);


  DeviceCapabilities(DeviceName, Port, DC_BINNAMES, PChar(BinNames), DevMode);





  for Loopy := 0 to NumSizes - 1 do


  begin


    FPaperSizeArr^[Loopy].Size := TWordArr(PaperSizes^)[Loopy];


    FPaperSizeArr^[Loopy].Dimensions := TPointArr(PaperDims^)[Loopy];


    if StrLen(PChar(PaperNames) + (Loopy * PaperSizeNames)) < PaperSizeNames then


      StrCopy(FPaperSizeArr^[Loopy].Name, PChar(PaperNames) + (Loopy * PaperSizeNames))


    else


      StrCopy(FPaperSizeArr^[Loopy].Name, 'Error');


  end;


  for Loopy := 0 to NumBins - 1 do


  begin


    FPaperBinsArr^[Loopy].Bin := TWordArr(Bins^)[Loopy];


    if StrLen(PChar(BinNames) + (Loopy * PaperBinsNames)) < PaperBinsNames then


      StrCopy(FPaperBinsArr^[Loopy].Name, PChar(BinNames) + (Loopy * PaperBinsNames))


    else


      StrCopy(FPaperBinsArr^[Loopy].Name, 'Error');


  end;


  FreeMem(BinNames, NumBins * PaperBinsNames * sizeof(char));


  FreeMem(Bins, NumBins * sizeof(Word));


  FreeMem(PaperNames, NumSizes * PaperSizeNames * sizeof(char));


  FreeMem(PaperDims, NumSizes * sizeof(TPOINT));


  FreeMem(PaperSizes, NumSizes * sizeof(Word));





  FreeLibrary(LibHandle);


end;





procedure TPrintPreview.SetFScale;


var


  ScaleX, ScaleY: Double;


  Temp: Double;


begin


  if (FScrollArea.Width > 16) and (FScrollArea.Height > 16) then


  begin


    {Check printer paper size against screen height/width}


    {Scale X and ScaleY are size of each screen pixel in mm}


    ScaleX := PageHiMet.X / FScrollArea.Width;


    ScaleY := PageHiMet.Y / FScrollArea.Height;


    {Scale is numer of printer pixels per screen pixel}


    if CurZoom = 1 then


    begin


      if ScaleY > ScaleX then


      begin


        {Picture fits vertically}


        FScale := PageHiMet.Y / ((FScrollArea.Height - 16));


      end


      else


      begin


        {Picture fits horizontally}


        FScale := PageHiMet.X / ((FScrollArea.Width - 16));


      end;


    end;


  end;


end;





procedure TPrintPreview.StartPreviewRender;


begin


  if Previewing and (FScale <> 0)then


  begin


    FPrintableArea.Canvas.Brush.Style := bsSolid;


    FPrintableArea.Canvas.Brush.Color := clWhite;


    FPrintableArea.Canvas.FillRect(Rect(0, 0, FPrintableArea.Width, FPrintableArea.Height));


    if Assigned(FOnRenderPage) then


    begin


      ChangeMapMode(FPrintableArea.Canvas, mmHiMetric, 0, 0);


      OnRenderPage(FCurrentPage, FPrintableArea.Canvas);


    end;


  end;


end;





procedure TPrintPreview.PrevPaint;


begin


  if FZooming then


  begin


    ValidateRect(FBackground.Canvas.Handle, nil);


  end


  else


  begin


    FBackground.Canvas.Brush.Bitmap := FBGBit;


    FBackground.Canvas.FillRect(Rect(0, 0, FBackground.Width, FBackground.Height));


    if FScale <> 0 then


    begin


      FBackground.Canvas.Pen.Color := clBlack;


      FBackground.Canvas.Pen.Style := psSolid;


      FBackground.Canvas.Brush.Color := clBlack;


      FBackground.Canvas.Brush.Style := bsSolid;


      FBackground.Canvas.Rectangle(FIndent.X + 2, FIndent.Y + 2,


                                   FIndent.X + 2 + Integer(Round(FPageHiMet.X / FScale)),


                                   FIndent.Y + 2 + Integer(Round(FPageHiMet.Y / FScale)));


      FBackground.Canvas.Brush.Color := clWhite;


      FBackground.Canvas.Rectangle(FIndent.X, FIndent.Y,


                                   FIndent.X + Integer(Round(FPageHiMet.X / FScale)),


                                   FIndent.Y + Integer(Round(FPageHiMet.Y / FScale)));


      FBackground.Canvas.Draw( FIndent.X + Integer(Round(FIndentHiMet.X / FScale)),


                               FIndent.Y + Integer(Round(FIndentHiMet.Y / FScale)),


                               FPrintableArea);


    end;


  end;


end;





function TPrintPreview.MouseToHiMet(MousePos: TPoint): TPoint;


begin


  if FScale <> 0 then


  begin


    MouseToHiMet.X := Integer(Round((MousePos.X - FIndent.X) * FScale));


    MouseToHiMet.Y := Integer(Round((MousePos.Y - FIndent.Y) * FScale));


  end


  else


  begin


    MouseToHiMet := Point(PageHiMet.X div 2, PageHiMet.Y div 2);


  end;


end;





procedure TPrintPreview.WMWinIniChange(var Message: TMessage);


begin


  if StrComp(PChar(Message.lParam), 'windows') = 0 then


    GetDefaultPrinterInfo;


  Resize;


  inherited;


end;





function TPrintPreview.GetFileName: String;


var


  OpenFilename: TOPENFILENAME;


  Filter: array[0..50] of char;


  Filename: array[0..128] of char;


  Title: array[0..15] of char;


begin


  StrPCopy(Filter, 'All files (*.*)' + Chr(0) + '*.*' + Chr(0) + Chr(0));


  StrPCopy(Filename, '');


  StrPCopy(Title, 'Print to file');


  OpenFilename.lStructSize := sizeof(TOPENFILENAME);


  OpenFilename.hWndOwner := Application.Handle;


  OpenFilename.hInstance := THandle(nil);


  OpenFilename.lpstrFilter := Filter;


  OpenFilename.lpstrCustomFilter := nil;


  OpenFilename.nMaxCustfilter := 0;


  OpenFilename.nFilterIndex := 0;


  OpenFilename.lpstrFile := Filename;


  OpenFilename.nMaxFile := 128;


  OpenFilename.lpstrFileTitle := nil;


  OpenFilename.nMaxFileTitle := 0;


  OpenFilename.lpstrInitialDir := nil;


  OpenFilename.lpstrTitle := Title;


  OpenFilename.Flags := OFN_NOREADONLYRETURN or OFN_OVERWRITEPROMPT or OFN_PATHMUSTEXIST;


  OpenFilename.nFileOffset := 0;


  OpenFilename.nFileExtension := 0;


  OpenFilename.lpstrDefExt := nil;


  OpenFilename.lCustData := 0;


  OpenFilename.lpfnHook := nil;


  OpenFilename.lpTemplateName := nil;


  if GetSaveFileName(OpenFilename) then


    GetFileName := StrPas(Filename)


  else


    GetFileName := '';


end;





procedure TPrintPreview.CreateDefaultAbortDlg(JobName: String);


begin


  DefaultAbortDlg := TDefaultAbortDlg.Create(Self);


  DefaultAbortDlg.Tag := 0;


  DefaultAbortDlg.DocName.Caption := JobName;


  if UseAbortDlgMode = dmGauge then


  begin


    DefaultAbortDlg.PageGauge.Visible := True;


    DefaultAbortDlg.PageLabel.Visible := False;


    DefaultAbortDlg.PageGauge.MinValue := FromPage;


    DefaultAbortDlg.PageGauge.MaxValue := ToPage;


    DefaultAbortDlg.PageGauge.Progress := FromPage;


  end


  else if UseAbortDlgMode = dmText then


  begin


    DefaultAbortDlg.PageGauge.Visible := False;


    DefaultAbortDlg.PageLabel.Visible := True;


    DefaultAbortDlg.PageLabel.Caption := 'Page: ' + IntToStr(FromPage);


  end;


  DefaultAbortDlg.Show;


end;





procedure TPrintPreview.EndDefaultAbortDlg;


begin


  DefaultAbortDlg.Free;


end;





function TPrintPreview.AbortProc(hdc: HDC; error: Integer): Boolean;


var


  Mesg: TMSG;


begin


  if Assigned(GlobalAbortProc) then


    GlobalAbortProc(GlobalPage);


  if DefaultAbortDlg.Visible then


  begin


    if DefaultAbortDlg.PageGauge.Visible then


      DefaultAbortDlg.PageGauge.Progress := GlobalPage


    else


      DefaultAbortDlg.PageLabel.Caption := 'Page: ' + IntToStr(GlobalPage);


  end;


  if DefaultAbortDlg.Tag <> 0 then


  begin


    GlobalAbort := True;


  end;





  while (not GlobalAbort) and PeekMessage(Mesg, THandle(nil), 0, 0, PM_REMOVE) do


  begin


    TranslateMessage(Mesg);


    DispatchMessage(Mesg);


  end;


  AbortProc := not GlobalAbort;


end;





procedure Register;


begin


  RegisterComponents('Samples', [TPrintPreview]);


end;





end.


